




REPORTING ROCK STARS OR ROBOTS, THEY WILL NEVER KNOW

Using SAS to Automate Report Distribution

PRESENTER: JACOB PRICE

Institutional Research & Testing

Baylor University



REPORTS

They want them.

We make them.

Repeat.



DELIVERY MODES

Welcome to Institutional Research and Testing


Welcome to the website of the Office of Institutional Research and Testing at Baylor University. Here you can find information about Institutional Research, Testing, Course Evaluations, and Survey Proposals at Baylor University.

The primary mission of the Office of Institutional Research is to conduct research in order to provide information which supports institutional planning, policy formulation, decision making, and external reporting. Our office also administers the faculty evaluations for Baylor University. We also offer various tests to enable students and members of the Waco community to achieve academic and professional goals through various testing programs.

Please feel free to [contact us](#) if you have any questions.

Fall 2016 Facts


- Baylor enrollment includes students from 50 states, the District of Columbia, and 86 foreign countries.
- The student to faculty ratio is 15 to 1. The average undergraduate class size is 27.
- Baylor's curriculum includes 142 baccalaureate programs, 75 master's programs, 42 doctoral programs, and the Education Specialist program.
- Library holdings include approximately 2.6 million bound volumes as well as numerous microforms, audiovisual, and electronic resources.
- As of May 31, 2016, the University endowment market value was approximately \$1.2 billion.
- Approximately 93% of Baylor undergraduate students receive student financial assistance.
- Baylor is one of the select 10% of U.S. colleges and universities with a Phi Beta Kappa



“

COULD YOU JUST EMAIL IT TO ME AS
AN ATTACHMENT?

”



THREE PIECES OF THE PROCESS

Contact Information

Customer Name

College

Department

Email

Etc.

Report

Data

Filtering

Arrangement

Analysis

Email

Body explaining purpose

Attachment

Signature

RETHINKING THE CONTACT LIST

	A	B	C	D	E	F
1	First Name	Last Name	College	Department	Level	Email
2	Georgia	Bellerd	MU	MUS	ALL	Georgia_Bellerd@baylor.edu
3	Dan	Berry	AS	GEO	GR	Dan_Berry@baylor.edu
4	Joel	Billings	ED	EDU	ALL	Joel_Billings@baylor.edu
5	Cecelia	Brown	SM	SEM	ALL	Cecelia_Brown@baylor.edu
6	Chelsea	Cane	BU	BUS	GR	Chelsea_Cane@baylor.edu
7	Ed	Carver	HN	HON	ALL	Ed_Carver@baylor.edu
8	Laurie	Freed	BU	OMBA	GR	Laurie_Freed@baylor.edu
9	Tamara	Garcia	AS	BIO	ALL	Tamara_Garcia@baylor.edu
10	Ryan	Garner	AS	BIO	GR	Ryan_Garner@baylor.edu
11	Rico	Guenad	IS	MED	ALL	Rico_Guenad@baylor.edu
12	Sharon	Gusukuna	AS	GEO	ALL	Sharon_Gusukuna@baylor.edu
13	Lindsey	Ham	ED	EDU	ALL	Lindsey_Ham@baylor.edu

- Add our Institutional Research & Testing (IRT) staff to the list.
- Add a signature column for standard employee email signatures.
- Add report flag columns to our contacts list.

Email	Signature	Course_Errors	Missing_Instructor	Records_Errors	Student
Marilyn_Hicks@baylor.edu				Y	Y
Stephanie_Hollis@baylor.edu			Y	Y	Y
Charles_King@baylor.edu		Y	Y		
Alfred_Lemon@baylor.edu					Y
Jerri_Maize@baylor.edu		Y	Y	Y	Y
Chava_McNair@baylor.edu		Y			
Mary_Pinscher@baylor.edu					Y
Nina_Porter@baylor.edu		Y			
Heather_Preen@baylor.edu		Y			
Jacob_Price@baylor.edu	 Jacob Price Institutional Research	Y	Y	Y	Y

IMPORT THE LIST INTO SAS

```
proc import out = present.contacts_list
    datafile = "k:\jacob price\contacts.xlsx"
    dbms = excel replace;
    range = "contacts";
    getnames = yes;
    mixed = no;
    scantext = yes;
    usedate = yes;
    scantime = yes;
run;
```

	First_Name	Last_Name	College	Department	Level	Email	Signature	Course_Errors	Missing_Instructor	Records_Errors	Student_Li
15	Marilyn	Hicks	SW	SWD	ALL	Marilyn_Hicks@baylor.edu				Y	Y
16	Stephanie	Hollis	NU	NUR	ALL	Stephanie_Hollis@baylor.edu			Y	Y	Y
17	Charles	King	AS	PSYN	ALL	Charles_King@baylor.edu		Y	Y		
18	Alfred	Lemon	HN	HON	ALL	Alfred_Lemon@baylor.edu					Y
19	Jerri	Maize	LW	LAW	ALL	Jerri_Maize@baylor.edu		Y	Y	Y	Y
20	Chava	McNair	AS	PHY	ALL	Chava_McNair@baylor.edu		Y			
21	Mary	Pinscher	BU	OMBA	GR	Mary_Pinscher@baylor.edu					Y
22	Nina	Porter	AS	PSYN	ALL	Nina_Porter@baylor.edu		Y			
23	Heather	Preen	AS	ENV	ALL	Heather_Preen@baylor.edu		Y			
24	Jacob	Price	IR	IRT	ALL	Jacob_Price@baylor.edu	 Jacob Price Institutional Research Baylor University One Bear Place #97032 Waco, TX	Y	Y	Y	Y
25	Lori	Puenney	GR	Graduate School	ALL	Lori_Puenney@baylor.edu				Y	Y
26	Thomas	Raker	HH	GTX	ALL	Thomas_Raker@baylor.edu			Y		Y
27	Paulette	Reinhaus	AS	GEO	ALL	Paulette_Reinhaus@baylor.edu		Y	Y		
28	Gina	Sankman	IS	BMS	ALL	Gina_Sankman@baylor.edu			Y		

THE MISSING INSTRUCTOR REPORT

- For this demo we will be creating the Missing Instructor Report from an already existing SAS data set, `missng_instructors`.

VIEWTABLE: Present.Missing_instructors							
	CRN	SECTION_NUMBER	COLLEGE	SUBJECT	COURSE_NUMBER	TITLE	ACTUAL_ENROLLMENT
1	31229	01	AS	ANT	3310	Intro Lang & Linguistics	3
2	36229	02	AS	ART	1300	Intro to Art (N-Major) AAI	141
3	19862	02	AS	ART	1310	Drawing I	15
4	35131	02	AS	ART	4100	Field Studies Art History AAI	9
5	32441	02	AS	ART	4321	Advanced Painting	2
6	36862	24	AS	BIO	1106	Mod Concp Bioscience Lab	28
7	12292	B	AS	BIO	3422	Human Physiology	23
8	12299	C	AS	BIO	3422	Human Physiology	23
9	12314	E	AS	BIO	3422	Human Physiology	24
10	11952	04	AS	CHE	1V98	Special Research Prob	2
11	11954	05	AS	CHE	1V98	Special Research Prob	1
12	11956	06	AS	CHE	1V98	Special Research Prob	2
13	11959	07	AS	CHE	1V98	Special Research Prob	1
14	22131	B	AS	CHE	4125	Biophysical Chemistry Lab	9
15	25462	01	AS	CHE	4141	Modern Biochemistry Laboratory	37
16	31310	F	AS	CHE	4142	Adv Biochemistry Laboratory	8
17	21184	02	AS	CHI	2320	Intermediate Chinese	11
18	12229	03	AS	CLA	3380	Classical Mythology	19
19	11185	03	AS	ENG	5125	British Lit & Cult	1

BEGINNING OUR MACRO

- First we begin our macro that we will call ***reports_and_emails***.
- Then we create a local macro variable called `&missing_count`.

- We use an `%IF` statement to move forward if there are records in the data set.

```
%macro reports_and_emails;
```

```
/* Count the number of sections with no instructor. */  
title1 'Current Number of Sections with no Instructor Assigned.';  
proc sql;  
    select nobs  
    into :missing_count  
    from sashelp.vtable  
    where libname = 'PRESENT'  
        and memname = 'MISSING_INSTRUCTORS'  
    ;  
quit;
```

```
/* If there are sections with no instructor, move forward. */  
%if &missing_count > 0  
    %then %do;
```

Current Number of Sections with no Instructor Assigned.

Number of Physical Observations
157

```
/* Sort the data for distribution. */  
proc sort data = present.missing_instructors  
  out = missing_instructors;  
  by college  
     subject  
     course_number  
     section_number;  
run;
```

```
/* Create a macro variable of the total number of  
colleges missing instructors. */  
data colleges;  
  set missing_instructors  
    (keep = college);  
  by college;  
  if first.college;  
    count + 1;  
run;
```

- The results viewer output for the &missing_count creation shows that there were observations in the data.
- We then create a copy of the data set sorted the way we need for our program with PROC SORT.
- Create a “colleges” data set with one record of each unique value in our data and add a count variable.

- We then create a &college_count macro variable that gives us the total number of Colleges with missing instructors.

```
title1 'Number of Colleges with Missing Instructors.';
proc sql;
  select max(count)
         into :college_count
  from colleges
  ;
quit;
```

Number of Colleges with Missing Instructors.

5

- The &college_count macro will be used to set the stop value of an index variable in an iterative %DO loop later in our code.

CREATING MACRO VARIABLES FOR CONTACTS

Internal Contacts (IRT)

- &irt_name – The first name of the contact in IRT that is sending the reports.
- &irt_email – Their email address.
- &irt_signature – Their email signature.

External Contacts (Recipients)

- &contact_name – The point of contact first name(s). There can be more than one person.
- &contact_email – The email address for the report recipient(s).
- &college – The College for which they are a point of contact.

IRT MACRO VARIABLES

- We use SELECT :INTO in PROC SQL to create our “irt” macro variables that will be used in our emails later.

```
proc sql noprint;
  select first_name,
         compress("'"||email||"'"),
         "'"||signature||'"
  into :irt_name,
       :irt_email,
       :irt_signature
  from present.contacts_list
  where department = 'IRT'
         and first_name = 'Jacob'
  ;
quit;
```

IRT_Name	IRT_Email	IRT_Signature
Jacob	'Jacob_Price@baylor.edu'	' Jacob Price Institutional Research Baylor University One Bear Place #97032 Waco, TX'

```
%do i = 1 %to &college_count.;
```

```
/* Set contact email and college macro values. */  
title1 'Contacts by College';  
proc sql;  
  select unique compress("'"||a.email||"'") 'Contact_Email',  
         a.college  
         format = $unit.  
  into:contact_email separated by ' ',  
       :college  
  from present.contacts_list a  
       inner join (missing_instructors b  
                  inner join colleges c  
                  on b.college = c.college)  
       on a.college = c.college  
  where missing_instructor = 'Y'  
         and c.count = &i.  
  ;  
quit;
```

Contacts by College

Contact_Email	College	Count
'Charles_King@baylor.edu'	AS	1
'Lisa_Willis@baylor.edu'	AS	1
'Paulette_Reinhaus@baylor.edu'	AS	1

CONTACT MACRO VARIABLES

- We begin an iterative %DO loop with an index value of &i using &college_count macro variable as the stop value.
- Using PROC SQL, our contacts' information is put into our "contact" macro variables.
 - By joining the "colleges" data set we can filter for only the records whose "count" value matches the current value of &i.

CREATING THE CUSTOM REPORT

- Name our report file.
- Open the ODS EXCEL dialog and set file options.
- Print the data to the file using PROC PRINT.
 - We are using &college in a WHERE statement to only pull records for the College that our contact belong fo.
- Close ODS EXCEL.

```
%let out = "K:\Jacob Price\Presentations\2017\Missing Instructors - &college..xlsx";
```

```
ods excel file = &out  
options (autofilter = 'on'  
        frozen_headers = 'yes'  
        embedded_titles = 'yes'  
        sheet_interval = 'bygroup'  
        sheet_name = '#byval(college)'  
        absolute_row_height = '14');
```

```
proc print data = missing_instructors  
noobs;  
by college;  
format college $unit.;  
var college  
    crn  
    subject  
    course_number  
    section_number  
    title  
    actual_enrollment;  
title1 'Courses With No Instructors';  
title2 "Send to Dean and Departmental Offices";  
where college = "&college.";  
run;
```

```
ods excel close;
```


REPORT OUTPUT

- Notice the report only contains courses for Arts and Sciences (AS), which is also the college the the contacts we created earlier were from.
- Also the tab on the spreadsheet has the College name on it.

College of Arts & Sciences

Courses With No Instructors
Send to Dean and Departmental Offices

COLL_CODE=College of Arts & Sciences

COLLEGE	CRN	SUBJECT	COURSE_NUMBER	SECTION_NUMBER	TITLE	ACTUAL_ENROLLMENT
College of Arts & Sciences	31229	ANT	3310	01	Intro Lang & Linguistics	3
College of Arts & Sciences	36229	ART	1300	02	Intro to Art (N-Major) AAI	141
College of Arts & Sciences	19862	ART	1310	02	Drawing I	15
College of Arts & Sciences	35131	ART	4100	02	Field Studies Art History AAI	9
College of Arts & Sciences	32441	ART	4321	02	Advanced Painting	2
College of Arts & Sciences	36862	BIO	1106	24	Mod Concp Bioscience Lab	28
College of Arts & Sciences	12292	BIO	3422	B	Human Physiology	23
College of Arts & Sciences	12299	BIO	3422	C	Human Physiology	23
College of Arts & Sciences	12314	BIO	3422	E	Human Physiology	24
College of Arts & Sciences	11952	CHE	1V98	04	Special Research Prob	2
College of Arts & Sciences	11954	CHE	1V98	05	Special Research Prob	1
College of Arts & Sciences	11956	CHE	1V98	06	Special Research Prob	2
College of Arts & Sciences	11959	CHE	1V98	07	Special Research Prob	1
College of Arts & Sciences	22131	CHE	4125	B	Biophysical Chemistry Lab	9
College of Arts & Sciences	25462	CHE	4141	01	Laboratory	37
College of Arts & Sciences	31310	CHE	4142	F	Adv Biochemistry Laboratory	8
College of Arts & Sciences	21184	CHI	2320	02	Intermediate Chinese	11
College of Arts & Sciences	12229	CLA	3380	03	Classical Mythology	19
College of Arts & Sciences	11195	CSS	5V35	02	Problems in Speech Comm	1
College of Arts & Sciences	29229	ENG	3305	01	Language in Society	12
College of Arts & Sciences	26771	ENG	4383	01	American Realism & Naturalism	12
College of Arts & Sciences	15279	ENV	1103	01	Wildlife Ecology Lab Exercises	21
College of Arts & Sciences	18400	ENV	3V90	08	Individual Research Problems	1
College of Arts & Sciences	36771	ENV	5102	01	Current Advances in Env Sci	5
College of Arts & Sciences	22971	FDM	4V85	06	Spec Prob Film/Digital Media	3
College of Arts & Sciences	23710	GEO	1401	A	Earthquakes-Natural Disasters	20
College of Arts & Sciences	25131	GEO	1402	J	World Oceans	18
College of Arts & Sciences	35184	GEO	3319	02	Intro to Geophysics	19
College of Arts & Sciences	21844	GEO	3430	01	Petrology	17
College of Arts & Sciences	21849	GEO	3430	A	Petrology	9
College of Arts & Sciences	11842	GEO	5V99	05	Thesis	1

College of Arts & Scienc +

DELIVERING THE REPORTS

- Set up the schedule.
 - Create a &today macro variable that contains the current day using the TODAY function.
- Then use a conditional %IF statement to determine whether or not to send the report today based on customer preferences.
 - For this program we used the day of the week and College as filters.

```
/* Set a macro value for the current day of the week. */  
data _null_;  
    call symput ('today', (put (today(), downname.)));  
run;
```

```
/* Send list on the days requested. */  
%if (&today = Thursday  
    and "&college" = "AS")  
or (&today = Tuesday)  
    %then %do;
```

SENDING THE EMAILS

- Use the FILENAME statement to create an external file that uses SMTP (Simple Mail Transfer Protocol).
 - We are using 'text/html' for the CONTENT_TYPE option because our email body and signature are in HTML format.
- Then use a data step with the _NULL_ keyword to create the body of the email and send it out.
 - Use PUT statements to insert text into the body of the email.

```
/* Email reports to IRT & appropriate contacts. */
filename mymail email from = 'Jacob_Price@baylor.edu'
  to = (&contact_email)
  cc = (&irt_email)
  subject = "Missing Instructors &college"
  content_type = 'text/html'
  attach = (&out.);

data _null_;
file mymail;
put '<font color = "003366" font face = "Californian FB"
  p style = "font-size:11pt">';
put "Good morning,";
put "<br> <br> Please see the attached list of courses that are missing";
put "instructors for the current term. This data is as of the end of";
put "the day yesterday. Census will be captured on the twelfth class day";
put "and, if possible, we need to record the correct faculty members for";
put "these courses. These instructor assignments directly impact course";
put "evaluations as well as instructor workload for a given term. Would";
put "you please reach out to your respective departments so that they can";
put "get these courses updated? Instructors can be added in ChairSIS. If";
put "you are no longer the correct contact for your area just let me";
put "know. Also, feel free to contact me if you have any questions.";
put "<br> <br> Thank you,";
put &irt_signature;
run;
```

Missing Instructors AS

To Charles_King@baylor.edu; Lisa_Willis@baylor.edu; Paulette_Reinhaus@baylor.edu

Cc Price, Jacob

Message

Missing Instructors - AS.xlsx (11 KB)

Good morning,

Please see the attached list of courses that are missing instructors for the current term. This data is as of the end of the day yesterday. Census will be captured on the twelfth class day and, if possible, we need to record the correct faculty members for these courses. These instructor assignments directly impact course evaluations as well as instructor workload for a given term. Would you please reach out to your respective departments so that they can get these courses updated? Instructors can be added in ChairSIS. If you are no longer the correct contact for your area just let me know. Also, feel free to contact me if you have any questions.

Thank you,

Jacob Price

Institutional Research

Baylor University

One Bear Place #97032

Waco, TX 76798

254.710.8836

"In God we trust; all others must bring data." – W. Edwards Deming

FINAL RESULT

We end up with a nicely formatted email, with the Missing Instructors Report for Arts and Sciences attached, that is sent to all of our Arts and Sciences contacts as well as our IRT contact.

We have also included all of the information the customer needs in the body of the email and the sender's email signature.

FINISHING UP OUR MACRO

- Close the %DO blocks.
 - The innermost %END statement closes the block that followed the last %IF condition –today being a scheduled delivery date.
 - The second %END statement completes the iterative %DO loop that set the value for &i and was used to create the &contact_email and &college macro variables.
 - The third and final %END statement closes out the code block for our first %IF condition – our initial data set having at least one observation

```
%end;
```

```
%end;
```

```
%end;
```

```
%mend reports_and_emails;
```

```
%reports_and_emails;
```

END THE MACRO

Finally we end our macro code with a `%MEND` statement and call the *reports_and_emails* macro to run it.



QUESTIONS?

REFERENCES

- **FILENAME Statement, EMAIL (SMTP) Access Method**
- SAS. 2016. SAS 9.4 Statements: Reference. Fifth edition. Chapter 2: Dictionary of SAS Statements, FILENAME Statement, EMAIL (SMTP) Access Method (pp. 112-127). Available at <http://support.sas.com/documentation/cdl/en/lestmtsref/69738/PDF/default/lestmtsref.pdf>
- **ODS EXCEL**
- Huff, Gina. 2016. "An 'Excel'ent Journey: Exploring the New ODS EXCEL Statement." *Proceedings of the 2016 SAS Global Forum Conference*. Available at <http://support.sas.com/resources/papers/proceedings16/2780-2016.pdf>
- SAS. 2015. SAS 9.4 Output Delivery System: User's Guide. Chapter 6: Dictionary of ODS Language Statements, ODS EXCEL Statement (pp. 292-320). Available at <http://support.sas.com/documentation/cdl/en/odsug/67921/PDF/default/odsug.pdf>
- **SAS Macro Language**
- Carpenter, Art. 2004. *Carpenter's Complete Guide to the SAS Macro Language*. Second edition. Cary, NC: SAS Institute Inc.

SAS CODE - 1

```

/*****
Last Submitted: 16FEB2017 - djp

Program name:  Presentation_Code

Location:     K:\Jacob Price\Presentations\TAIR\2017

Purpose:     To demonstrate concepts for presentation.

Keywords:    macros, contact list, email

Modifications to program:

15FEB2017    Created program. - djp

Inputs:      census data

Outputs:     spreadsheets, emails

Updates:

Special Instructions:
*****/

/* Librefs. */
libname present 'k:\jacob price\presentations\2017';

options mlogic
        minoperator;

/*****/
/*          Create Contacts List Dataset          */
/*****/

proc import out = present.contacts_list
            datafile = "k:\jacob price\contacts.xlsx"
            dbms = excel replace;
            range = "contacts";
            getnames = yes;
            mixed = no;
            scantext = yes;
            usedate = yes;
            scantime = yes;
run;
```

SAS CODE - 2

```
/* *****  
/*          Prepare the Missing Instructor Data Reports          */  
/* *****  
  
%macro reports_and_emails;  
  
/* Count the number of sections with no instructor. */  
title1 'Current Number of Sections with no Instructor Assigned.';  
proc sql;  
    select nobs  
    into :missing_count  
    from sashelp.vtable  
    where libname = 'PRESENT'  
          and memname = 'MISSING_INSTRUCTORS'  
    ;  
quit;  
  
/* If there are sections with no instructor, move forward. */  
%if &missing_count > 0  
    %then %do;  
  
    /* Sort the data for distribution. */  
    proc sort data = present.missing_instructors  
        out = missing_instructors;  
        by college  
           subject  
           course_number  
           section_number;  
    run;  
  
    /* Create a macro variable of the total number of  
    colleges missing instructors. */  
    data colleges;  
        set missing_instructors  
            (keep = college);  
        by college;  
        if first.college;  
        count + 1;  
    run;
```

SAS CODE - 3

```
title1 'Number of Colleges with Missing Instructors.';
proc sql;
  select max(count)
         into :college_count
         from colleges
        ;
quit;

proc sql noprint;
  select first_name,
         compress("'"||email||"'"),
         "'"||signature||'"
  into :irt_name,
       :irt_email,
       :irt_signature
  from present.contacts_list
  where department = 'IRT'
         and first_name = 'Jacob'
        ;
quit;

%do i = 1 %to &college_count.;

/* Set contact email and college macro values. */
title1 'Contacts by College';
proc sql;
  select unique compress("'"||a.email||"'") 'Contact_Email',
         a.college,
         c.count 'Count'
  into:contact_email separated by ' ',
       :college
  from present.contacts_list a
       inner join (missing_instructors b
                  inner join colleges c
                  on b.college = c.college)
       on a.college = c.college
  where missing_instructor = 'Y'
         and c.count = &i.
        ;
quit;

%let out = "K:\Jacob Price\Presentations\2017\Missing Instructors - &college..xlsx";
```

SAS CODE - 4

```
ods excel file = &out
  options (autofilter = 'on'
         frozen_headers = 'yes'
         embedded_titles = 'yes'
         sheet_interval = 'bygroup'
         sheet_name = '#byval(college)'
         absolute_row_height = '14');

proc sort data = missing_instructors;
  by college
     subject
     course_number
     section_number;

run;

proc print data = missing_instructors
  noobs;
  by college;
  format college $unit.;
  var college
      crn
      subject
      course_number
      section_number
      title
      actual_enrollment;
  title1 'Courses With No Instructors';
  title2 "Send to Dean and Departmental Offices";
  where college = "&college.";
run;

ods excel close;

/* Set a macro value for the current day of the week. */
data _null_;
  call symput('today', (put(today(), downname.)));
run;

/* Send list on the days requested. */
%if (&today = Friday
    and "&college" = "AS")
or (&today = Tuesday)
  %then %do;
```

SAS CODE - 5

```
/* Send list on the days requested. */
%if (&today = Friday
    and "&college" = "AS")
or (&today = Tuesday)
    %then %do;

    /* Email reports to IRT & appropriate contacts. */
    filename mymail email from = 'Jacob_Price@baylor.edu'
        to = (&contact_email)
        cc = (&irt_email)
        subject = "Missing Instructors &college"
        content_type = 'text/html'
        attach = (&out.);

    data _null_;
    file mymail;
    put '<font color = "003366" font face = "Californian FB"
        p style = "font-size:11pt">';
    put "Good morning,";
    put "<br> <br> Please see the attached list of courses that";
    put "are missing instructors for the current term. This data";
    put "is as of the end of the day yesterday. Census will be";
    put "captured on the twelfth class day and, if possible, we";
    put "need to record the correct faculty members for these";
    put "courses. These instructor assignments directly impact";
    put "course evaluations as well as instructor workload for a";
    put "given term. Would you please reach out to your";
    put "respective departments so that they can get these";
    put "courses updated? Instructors can be added in ChairSIS.";
    put "If you are no longer the correct contact for your area";
    put "just let me know. Also, feel free to contact me if you";
    put "have any questions.";
    put "<br> <br> Thank you,";
    put &irt_signature;
    run;

    %end;

%end;

%end;

%mend reports_and_emails;

%reports_and_emails;
```


REPORTING ROCK STARS OR ROBOTS, THEY WILL NEVER KNOW

Using SAS to Automate Report Distribution

Presenter – Jacob Price
jacob_price@baylor.edu
Baylor University